
News  

The CED Forum is now live! 
Our website now includes a user forum and bulletin board section where registered 
members can post questions about software and hardware for the CED team, 
suggestions for future products and software revisions and browse generally useful 
information posted by ourselves and other CED users.  

You can browse the forum as a guest but to submit posts you need to register as a 
member here. We have tried to make the registration process as simple as possible while 
providing protection against spamming and only require the bare minimum of information 
(email addresses will be kept strictly private, of course).  

Meetings and events 
FENS (6th Forum of European Neuroscience) 
Palexpo Conference Center 
Geneva, Switzerland  
July 13th – 16th 2008 

Physiological Society Main Meeting 
University of Cambridge 
Cambridge, UK  
July 13th – 16th 2008 

5th Symposium for Chinese Neuroscientists Worldwide 
Changsha City 
Hunan Province, China  
July 25th – August 1st 2008 

Scandinavian Physiological Society Annual Meeting 
University of Oulu 
Oulu, Finland 
August 15th – 17th 2008 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.09 Signal version 4.02 

Spike2 version 5.20 Signal version 3.10 

Spike2 demo Signal demo 

http://www.ced.co.uk/phpBB3/ucp.php?mode=register&sid=61e53f187c0c046b19fc573aab1085ee
http://fens2008.neurosciences.asso.fr/
http://www.physiology2008.org/
http://www.csn.org.cn/news.asp?id=160
http://www.oulu.fi/sps2008/
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigdemu.shtml


Spike2   
 

 

Q. Is there any way to generate a display of overdrawn data sections from a Spike2 
file without using a script? 

A. From Spike2 version 6.09 onwards you can overdraw time view data on and off-
line to compare responses to stimuli and optionally show these overdrawn 
sections in 3D as a ‘waterfall’ type display. As overdraw is only a display mode, 
all measurements and cursor positions will behave as if the overdrawn data is not 
present. Overdraw mode is available from the Display Trigger dialog, which can 
be found in the Spike2 View menu Trigger/Overdraw function. 

 The Display Trigger provides an oscilloscope-type display of time view data with 
an optional pre-trigger time. It runs as a paged display or updates in response to 
events on a trigger channel. Pre-trigger and minimum hold display times can be 
set from the dialog, as well as options for Cursor 0 behaviour. For an example of 
using the display trigger, see eNewsletter #2.  

 
Display trigger dialog with overdraw enabled 

 To enable the overdraw section of the dialog, both the Enable trigger and Zero x 
axis at trigger options must be checked. With Enable overdraw selected, data 
sections are drawn over each other in time order according to the selected 
triggers, with the newest section of data, or frame, at the front of the display. 
There are options for limiting the number of frames displayed and the time span 
of the frame sections as well as settings to determine the colouring of overdrawn 
frame data. Trigger times can be also added to the display using the Overdraw 
List dialog. 

 

 
LTP data shown in overdraw triggered display mode 

CED
Overdraw mode can be used on-line and off-line. You can use it with the View menu Rerun function to ‘replay’ your data for presentations



 The Enable 3D display option draws each frame of data with a vertical and 
horizontal offset to give a waterfall display effect. The screen arrangement for the 
channels and frames can be controlled from the associated Overdraw 3D dialog. 

  
3D overdrawn data on multiple channels with displayed 3D settings 

 Full details of the overdraw mode can be found in the Spike2 on-line help.  

  
 

Q. Is it possible to export data from Signal to MATLAB compatible files? 

A. You can export data files and XY view data to MATLAB from the Export as… 
function in the Signal File menu. You must select the MATLAB support option 
during the Signal 4.02 installation to enable this feature. This can be done by 
reinstalling the software if you have not previously selected this option. 

 Associated dialogs allow you to select the channels, frames and time range to 
export as well as the required compatibility and data options for the .mat file 
created. 

  
Export dialog settings for Signal data 

 As well as support for the export of Signal data, there are new script commands 
to move values between Signal and MATLAB variables and execute arbitrary 
commands in the MATLAB environment. Unlike the export function, these script 
commands can only be used on systems that have a licensed copy of MATLAB 
installed. 

Scripts: Spike2  
Q. Is it possible to get information from a sampling configuration as a text file? 

Something like a report file holding information such as number and type of 
channels, sample rates and any associated 1902 settings, for example? 

A. The attached script, PrintMyConfig.s2s, prints information stored in the 
current sampling configuration to a text file. The script creates a dialog from 

Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

CED
You can export error information along with the waveform values by selecting Waveforms and errors from the Layout options in the MATLAB export dialog


'$PrintMyConfig.s2s|Print contents of current sampling configuration
'===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB6 0FE, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design. last modified: July 2008.
'Author:	GH.

'SOFTWARE REQUIRED:
'Spike 2 version 6.09 or higher
'The script makes use of script functions contained in the file GHutils.s2s
'This additional script is located in a folder called <include> inside the Spike2 folder.

'HEALTH WARNING:
'The script is a <work in progress> and is offered without guarantees.

'OVERVIEW
'This script allows you to print out most of the information stored in the current sampling configuration and 
'the current settings of any signal conditioners.
'We hope that the printout will be useful for your records, -perhaps an appendix to a Ph.D thesis.

'USER GUIDE
'Keep this script in the <scripts> folder of the <Spikex> directory (where x stands for the version number).
'The script needs to open a data file for sampling in order to extract the required information. Therefore, the script will 
'only work if your 1401 interface is connected, switched on and is capable of running with this sampling configuration.
'Similarly, information about Signal conditioners will only be available if they are switched on at the time.
'Note that the report gives the most recent setup of the signal conditioners -which is not necessarily the same as that used
'when a previous data file was recorded.

'When you run this script for the first time, it will install a button <PrintCfg> on the Script Bar. This button
'will give you "one click" access to the script in future.

'Before running the script, make sure that the relevant sampling configuration is loaded. You could do this via the Load Configuration
'option on the Spike2 File menu or by clicking the relevant button on the Sample Bar (if you created one). If a named sampling configuration
'was not explicitly loaded or if the current configuration was modified during sampling, it will be renamed automatically to <Last.s2c>.
'See the Spike2 Help for further information.

'When you run the script, a dialog will allow you to select the components of the configuration to print via check boxes.
'The options are:

'GENERAL:
'Lists todays date, Spike 2 version and serial number, current operating system, number of monitors and their coordinates.
'It also indicates the model of 1401 being used, and the temporary location of data files during sampling.

'CHANNELS
'This tabulates the data shown in the Channels section of the Sampling configuration dialog.
'It also shows information on:
'	channels that are created online such as Virtual channels
'	Event Marker and RealMark channels created online using Active cursors and the Spike2 Measuremeants facility.
'	Duplicate channels created online.

'RESOLUTION, SAMPLING MODE, PLAY WAVEFORM and AUTOMATION
'Checking these boxes will tabulate the data visible in the corresponding tab of the Sampling onfiguration dialog.

'OUTPUT SEQUENCER
'This gives the name and path of a text sequence file if one was used and the method by which the sequence is controlled.
'However, it does not provide any information on sequences generated by the graphical sequence editor.

'ACTIVE CURSORS
'This provides a listing of the parameters used by online active cursors to generate online measurement channels. The contents
'of each column in the table depend somewhat on the type of search. (See the Spike2 Help on the script commands used to retrieve
'the information (CursorActiveGet and CursorActiveSet) for further details. Note also that active cursor expressions are abbreviated
'for better tabulation. For example Cursor(0) is shortened to C0 and MaxTime() to MaxT.

'CHANNEL PROCESSES
'This identifies channels with channel processes and list them together with their parameters

'SIGNAL CONDITIONERS
'Three types of Signal conditioners are are recognised. the CED 1902, Axon CyberAmp and CED Power 1401 variable gain. In the latter case,
'only the gain setting is listed. In the case of CyberAmps and 1902s, the table lists:
'Port, Channel, Channel title, Gain, Offset, Input and Filter settings.
'Some models of 1902 have additional options such as Rectification. These features are also documented if selected.

'A text file containing the information you requested will display when you click on OK in the Set up dialog.
'You can save the information as a text file via <Save As> on the Spike2 File menu.
'Alternatively, you could <Select All> and Copy/Paste the text to a spreadsheet via the clipboard.

#include <GHutils.s2s>	                               'include functions stored in include\GHutils.s2s (inside the Spike2 folder)
var tvh%,txtvh%,chk%[10],ok%;
var s2home$,set$,cfgname$;
var DAndx%;	                                           'index of 1401 in use.
s2home$:=FilePath$(2);

set$:=Print$("PrintCfg|%s\scripts\\PrintMyConfig.s2s|Print current sampling configuration",s2home$);
ok%:=AddScriptBarBtn%(set$);
if ok%<0 then
	Newsflash("Failed to add a button to the Script Bar. No space available.",5.0,0,10,1);
endif;
View(App(3)).WindowVisible(2);	                      'iconise this script
ok%:=Scope%();	                                        'selects which bits to print
if ok%<>1 or ArrSum(chk%)=0 then	                      'cancelled or nothing selected
	halt	                                              'cancelled
endif;
HideApps();
DAndx%:=U1401Open();	                                  'get index of model of 1401 in use
U1401Close();

tvh%:=FileNew(0,2);	                                  'mode% 2: include additional measurement channels and virtual channels created online
if tvh% < 0 then
	Newsflash("Please switch on the 1401 and retry.",10.0,0,10,1);
	halt
endif;

txtvh%:=FileNew(1);	                                  'open text file
FontSet(1,9,0);	                                     'courier-like font; point size: 9
TabSettings(11);
if chk%[0] then
	General%();
endif;

cfgname$:=SampleConfig$();	                            'show title irrespective of check box status
if cfgname$="" then	                                  'sampling config updated by sampling so previous name is no longer valid
	cfgname$:="Last.s2c\n";
endif;
Print("\nCURRENT SAMPLING CONFIGURATION:\t%s\n",cfgname$);

if chk%[1] then
	ChannelsTab%();
endif;
if chk%[7] then
	ActiveCursors%();
endif;
if chk%[8] then
	ChannelProcesses%()
endif;
if chk%[9] then
	Conditioners%();
endif;
if chk%[2] then
	ResolutionTab%();
endif;
if chk%[3] then
	ModeTab%();
endif;
if chk%[4] then
	SequencerTab%();
endif;
if chk%[5] then
	PlaywaveTab%();
endif;
if chk%[6] then
	Automation%();
endif;

View(tvh%);
FileClose(1,-1);	                                     'close windows associated with this time view (doesn-t work for XY and result views)
CloseViews(16+4096);	                                  'close XY and result views created by config
RestoreApps();
View(txtvh%);
Window(0,0,100,99);
Draw(1);	                                              'show line 1 at top of screen
FrontView(txtvh%);
halt

proc CloseViews(kind%)	                               'close empty xy and result views created by sampling configuration
var lst%[2],nvws%,i%,j%,n%,pts%;

nvws%:=ViewList(lst%[],kind%);	                      'check whether any views were created
if nvws%>0 then
	var vlst%[nvws%+1];
	ViewList(vlst%[],kind%);	                         'list of xy views
	for i%:=1 to vlst%[0] do
		View(vlst%[i%]);
		docase
		case ViewKind()=12 then	                         'xy view
			if ChanList(lst%[])> 1 or XYCount(1) > 0 then	'this xy view has some data in it;
				'do nothing.
			else			
				FileClose(0,-1);	                         'empty xy view; discard it without compunction
			endif;
			
		case ViewKind()=4 then	                         'result view
			n%:=ChanList(lst%[],1);	                      'count channels
			for j%:=1 to n% do
				pts%+=Count(j%,0,MaxTime());	             'total of data points
			next;
			if pts%=0 then
				FileClose(0,-1);	                         'delete empty result view;
			else
				pts%:=0;
			endif;
		endcase;
	next;
endif;
return;
end;

func Scope%();	                                        'select the scope of the printout
var i%,ok%,pr$[10],x$,y$;
var chkstate$;
var xpos,ypos;

pr$[0]:="General"; pr$[1]:="Channels"; pr$[2]:="Resolution"; pr$[3]:="Mode"; 
pr$[4]:="Sequencer"; pr$[5]:="Play waveform"; pr$[6]:="Automation";
pr$[7]:="Active Cursors"; pr$[8]:="Channel Processes"; pr$[9]:="Conditioner Settings";

Profile("PrintConfig","dlgx","0",x$);	                'get last coordinates of the dialog from registry
Profile("PrintConfig","dlgy","0",y$);
Profile("PrintConfig","chkstate","1|1|1|1|1|1|1|1|1|1",chkstate$);	'extract previous check box settings from registry
ReadSetup("|");
ReadStr(chkstate$,chk%[0],chk%[1],chk%[2],chk%[3],chk%[4],chk%[5],chk%[6],chk%[7],chk%[8],chk%[9]);

DlgCreate("Scope of the print out",Val(x$),Val(y$));
DlgGroup("",2,0.6,42,6);
xpos:=5; ypos:=1.3;
for i%:=0 to 4 do
	DlgCheck(i%+1,pr$[i%],xpos,ypos);
	ypos+=1;
next;
xpos:=24; ypos:=1.3;
for i%:=5 to 9 do
	DlgCheck(i%+1,pr$[i%],xpos,ypos);
	ypos+=1;
next;
DlgButton(3,"All On",AllOn%);
DlgButton(2,"All Off ",AllOff%);
DlgButton(1,"OK|hotkey: Enter|0x0d",MyOK%);
DlgButton(0,"Cancel|hotkey: Space bar|0x20",MyCancel%);	'shortcut: spacebar
DlgAllow(511,0,ScopeCh%);
ok%:=DlgShow(chk%[]);
if ok%=1 then
	chkstate$:=Print$("%d|%d|%d|%d|%d|%d|%d|%d|%d|%d",chk%[0],chk%[1],chk%[2],chk%[3],chk%[4],chk%[5],chk%[6],chk%[7],chk%[8],chk%[9]);
	Profile("PrintConfig","chkstate",chkstate$);	       'store current check box settings in the registry
endif;
return ok%;
end;

func ScopeCh%(item%)	                                  'change function of the Scope dialog
var i%,onoff%[10],en%:=1;

for i%:=1 to 10 do
	onoff%[i%-1]:= DlgValue(i%);	                      'make array of initial checkbox states
next;
if ArrSum(onoff%[])=0 then
	en%:=0;
endif;
DlgEnable(en%,-1);	                                  'disable OK button if no boxes are checked
return 1;
end;

func AllOff%()	                                        'disable all check boxes
var i%;
for i%:=1 to 10 do
	DlgValue(i%,0);
next;
DlgEnable(0,-1);	                                     'disable OK
return 1;
end;

func AllOn%()	                                        'enable all check boxes
var i%;
for i%:=1 to 10 do
	DlgValue(i%,1);
next;
DlgEnable(1,-1);	                                     'enable OK
return 1;
end;

func MyCancel%()	                                     'custom Cancel button function
SaveDlgCoords();	                                     'save the last coordinates of the dialog for next time
return 0;
end;

func MyOK%()	                                        'custom OK button function
SaveDlgCoords();	                                     'save the last coordinates of the dialog for next time
return 0;
end;

proc SaveDlgCoords()	                                  'save the last coordinates of the dialog for next time
var dlgx,dlgy;	                                        'coordinates of scope dialog

DlgGetPos(dlgx,dlgy,-1);
Profile("PrintConfig","dlgx",Print$("%.1f",dlgx));	    'save last coordinates of the dialog to registry for next time
Profile("PrintConfig","dlgy",Print$("%.1f",dlgy));
return;
end;

func General%()	                                     'based on code from SG
Var ver,system%,pr$,nmon%,i%;
var evtsndx%,VR1401ndx%,newdatadir$;
var VRange$[3],evtsat$[2];
VRange$[0]:="5 V"; VRange$[1]:="10 V"; VRange$[2]:="same as last 1401";
evtsat$[0]:="front panel"; evtsat$[1]:="rear connector";
var DA$[6];	                                           'model of 1401 in use
DA$[0]:="Standard 1401"; DA$[1]:="1401plus"; DA$[2]:="Micro1401"; 
DA$[3]:="Power1401"; DA$[4]:="Micro1401 mkII"; DA$[5]:="Power1401 MkII"; 

pr$:="Operating system:\tWindows";

View(txtvh%);
Print("GENERAL INFORMATION\n");
Print("%s\n",Date$());	                               'Print the date
ver:=App(-1);
ver/=100;
Print("Spike2 version nr:\t%.2f\n", ver);	             'Print the version of Spike2 
Print("Spike2 serial nr:\t%d\n", App(-3));	          'and the serial number

System%:=System();	                                  'Get the operating system we are using
docase	                                              'Print the OS
case System%=351 then
	Print("%s NT 3.51\n",pr$);
case System%=400 then
	Print("%s 95 or NT\n",pr$);
case System%=410 then
	Print("%s 98\n",pr$);
case System%=490 then
	Print("%s Me\n",pr$);
case System%=500 then
	Print("%s NT 2000\n",pr$);
case System%=501 then
	Print("%s XP\n",pr$);
case System%=502 then
	Print("%s XP x64 Edition\n",pr$);
case System%=600 then
	Print("%s Vista\n",pr$);
endcase;

Print("\nD/A INTERFACE\n");
Print("%s\n",DA$[DAndx%]);

nmon%:=System(1,0);
Print("\nMONITORS\t%d\n",nmon%);
var sz%[4];	                                           'pixel coordinates of each monitor
Print("Screen Coordinates:\n\txlow\tylow\txhi\tyhi\n");
for i%:=1 to nmon% do
	System(1,i%,sz%[]);
	Print("Screen %d:\t%4d\t%4d\t%4d\t%4d\n",i%,sz%[0],sz%[1],sz%[2],sz%[3]);
next;

Profile("Preferences","Event ports at rear",0,evtsndx%);	'0: evts 0 and 1 on front panel; 1: on rear connector;
Profile("Preferences","Ten volt 1401",0,VR1401ndx%);	 'voltage range of 1401; 0: 5V; 1: 10V; 2: same as last 1401.
Profile("Preferences","New file path","",newdatadir$);	'temporary location for data files during sampling
if newdatadir$="" then newdatadir$:="current folder"; endif;

Print("\nSAMPLING PREFERENCES\n");
if DAndx%>=2 then
	Print("Event ports 0 and 1 on:  %s\n",evtsat$[evtsndx%]);	'option for Power and Micro-1401 only
endif;
Print("1401 voltage range:  %s\n",VRange$[VR1401ndx%]);
Print("Temporary location for data files during sampling:  %s\n",newdatadir$);
return 1;
end;

func ChannelsTab%()
var kind%;
var rateok$;
var nch%;
var a$;
var i%,c%;

var lst%[2];
var npts%,pre%,ntr%,port$;	                            'Wavemark trace parameters
var drvch%[400];	                                     'list of derived merker and event channels stored (invisibly) in config
var vrmch%[200];	                                     'list of derived virtual and RealMark channels stored in config
var type$[10];	                                        'element [0] unused
type$[1]:="Waveform"; type$[2]:="Event -"; type$[3]:="Event +"; type$[4]:="Level"; 
type$[5]:="Marker"; type$[6]:="WaveMark"; type$[8]:="TextMark"; type$[7]:="RealMark"; type$[9]:="RealWave";

View(tvh%);
nch%:=ChanList(lst%,16384);	                         'count all channels except duplicates
var chlst%[nch%+1];	                                  'create array big enough to hold them
ChanList(chlst%);	                                     'get list
View(txtvh%);
Print("SAMPLED CHANNELS\n");
Print("Chan\tType\tPort\tTitle\tEvt rate\tADC rate\tOK\tScale\tOffset\tUnits\tPts\tPre\tComment\n");
View(tvh%);
for i%:=1 to chlst%[0] do
	c%:=chlst%[i%];
	kind%:=ChanKind(c%);
	docase
	case kind%=1 then	                                  'waveform channel
		if SampleChanInfo(c%,4)=SampleChanInfo(c%,3) then rateok$:=" Y "; else rateok$:=" N "; endif;	'indicate whether actual adc rate matches idealised
		a$:=Print$("%3d\t%s\t%3d\t%s\t\t%.2f\t%s\t%.3f\t% .3f\t %s\t\t\t%s\n",c%,type$[kind%],SampleChanInfo(c%,1),ChanTitle$(c%),SampleChanInfo(c%,4),rateok$,ChanScale(c%),ChanOffset(c%),ChanUnits$(c%),ChanComment$(c%));
		
	case (kind%> 1 and kind% <=5) or kind%=8 then	    'events and markers including TextMarks
		docase
		case c%>=30 and c%<=32 then	                   'special channel with no port number
			a$:=Print$("%3d\t%s\t\t%s\t%3.0f\t\t\t\t\t\t\t\t%s\n",c%,type$[kind%],ChanTitle$(c%),SampleChanInfo(c%,2),ChanComment$(c%));
		case SampleChanInfo(c%,1)>=0 then	             'channel with a valid port number, ergo: part of the config and not a derived channel
			a$:=Print$("%3d\t%s\t%3d\t%s\t%3.0f\t\t\t\t\t\t\t\t%s\n",c%,type$[kind%],SampleChanInfo(c%,1),ChanTitle$(c%),SampleChanInfo(c%,2),ChanComment$(c%));
		else
			drvch%[0]+=1;	                               'make a channel list of derived event and marker channels
			drvch%[drvch%[0]]:=c%;
		endcase;
		
	case kind%=6 then	                                  'WaveMark
		if SampleChanInfo(c%,4)=SampleChanInfo(c%,3) then rateok$:=" Y "; else rateok$:=" N "; endif;	'indicate whether actual adc rate matches idealised
		npts%:=MarkInfo(c%,pre%,ntr%);
		if ntr%>1 then
			port$:=Print$("%d-%d",SampleChanInfo(c%,1),SampleChanInfo(c%,1)+ntr%-1);	'cope with multiple traces in port listing
		else
			port$:=Print$("%d",SampleChanInfo(c%,1));
		endif;		
		a$:=Print$("%3d\t%s\t %s\t%s\t% .0f\t% .0f\t%s\t% .3f\t% .3f\t %s\t% d\t% d\t%s\n",c%,type$[kind%],port$,ChanTitle$(c%),SampleChanInfo(c%,2),SampleChanInfo(c%,4),rateok$,ChanScale(c%),ChanOffset(c%),ChanUnits$(c%),npts%,pre%,ChanComment$(c%));		
	endcase;	
	View(txtvh%).Print("%s",a$);	                      'add line to document
	a$:="";
next;

nch%:=ChanList(lst%,64+512);	                         'count realmarks and realwaves
if nch%>0 then
	ChanList(vrmch%[],64+512);	                         'combine lists of derived event realmark and virtual channels
	ArrConst(drvch%[drvch%[0]+1:],vrmch%[1:]);
	drvch%[0]+=nch%;	                                  'update count
	ArrSort(drvch%[1:drvch%[0]]);	                      'ascending order
endif;
if drvch%[0]>0 then
	var ret[3],expr$,j%;	                               'virtual channel parameters
	View(txtvh%);
	Print("\nONLINE MEASUREMENT CHANNELS\n");
	Print("Chan\tType\tTitle\tUnits\tMatch\tIntvl(s)\tAlign(s)\tExpr.\tComment\n");
	View(tvh%);
	for i%:=1 to drvch%[0] do
		c%:=drvch%[i%];
		kind%:=ChanKind(c%);
		a$:=Print$("%3d\t%s\t%s\t %s\t",c%,type$[kind%],ChanTitle$(c%),ChanUnits$(c%));
		if kind% = 9 then	                               'virtual channel created online
			for j%:=1 to 3 do
				ret[j%-1]:=VirtualChan(c%,j%,expr$);	    'get virtual channel info
			next;
			a$+=Print$("% .0f\t%.4g\t%.4g\t%s\t",ret[0],ret[1],ret[2],expr$);
		else
			a$+="  -\t  -\t  -\t  -\t";	                'dashes for virtual channel related items
		endif;
		a$+=Print$("%s\n",ChanComment$(c%));	          'comment at end of line because length is so variable
		View(txtvh%).Print("%s",a$);	                   'add line to document
		a$:="";		
	next;
endif;

nch%:=ChanList(lst%[],4096+8192+2097152);	             'duplicate channels only
if nch% > 0 then
	View(txtvh%).Print("\nDUPLICATE CHANNELS\nChan\tType\tTitle\tUnits\tComment\n");
	var dchlst%[nch%+1];
	ChanList(dchlst%[],4096+8192+2097152);	             'fill list with duplicate channels
	for i%:=1 to nch% do
		kind%:=ChanKind(dchlst%[i%]);
		a$:=Print$("%3d\t%s\t%s\t%s\t%s",dchlst%[i%],type$[kind%],ChanTitle$(dchlst%[i%]),ChanUnits$(dchlst%[i%]),ChanComment$(dchlst%[i%]));
		View(txtvh%).Print("%s\n",a$);
	next;
endif;
return 1;
end;

func ResolutionTab%()
var i%;
var tmax,hr%,min%,sec%;	                               'max file 
var resinfo%;
var UsPerT,TPerAdc;
var resinfo[7];	                                     'current resolution settings
var pr0$[3],pr1$[3],pr2$[8];	                         'SampleOptimise info

pr0$[0]:="None: use manual settings";
pr0$[1]:="Partial: fixed time units";
pr0$[2]:="Full: time units can change";

pr1$[0]:="Version 3 compatible";
pr1$[1]:="Keep same sample rate groups";
pr1$[2]:="Ignore same sample rate groups";

pr2$[0]:="General compatibility";
pr2$[1]:="1401plus with old ADC";
pr2$[2]:="Power1401";
pr2$[3]:="1401plus or a micro1401.";
pr2$[4]:="Micro1401 mk II";
pr2$[5]:="Power1401 625";
pr2$[6]:="micro1401 (not mk II)";
pr2$[7]:="Power1401 mk II";

UsPerT:=SampleUsPerTime();
TPerAdc:=SampleTimePerAdc();
tmax:=(Pow(2,31)-1)*UsPerT*1e-6;	                      'max sampling time in seconds
HrMinSec(tmax,hr%,min%,sec%);
for i%:=0 to 6 do
	resinfo[i%]:=SampleOptimise(i%);
next;
View(txtvh%);
Print("\nRESOLUTION\n");
Print("Microseconds per time unit \t%.1f uS\n",UsPerT);
Print("Time units per ADC convert \t%d\n",TPerAdc);
Print("Longest run time (h:m:s)\t%2d : %02d : %02d\n",hr%,min%,sec%);
Print("ADC will convert at:\t%.1f Hz\n\n",1e6/(TPerAdc*UsPerT));
Print("Optimise:\t%s\n",pr0$[resinfo[0]]);	          'optimisation of waveform rates
Print("Groups:\t%s\n",pr1$[resinfo[1]]);	             'use of quick channels and channel groups
Print("1401:\t%s\n\n",pr2$[resinfo[2]]);	             '1401 compatibility
if resinfo[0]=2 then
	Print("Low limit for microseconds per time:\t %3d uS\n", resinfo[3]);	'Low and high limits for optimisation
	Print("High limit for microseconds per time:\t %3d uS\n", resinfo[4]);	'only relevant in full optimisation selected
endif;
resinfo%:=resinfo[5];
if resinfo% band 1 =1 then
	Print("10 MHz clock disabled.\n")
endif;
if resinfo% band 2 =2 then
	Print("WaveMark copy disabled.\n")
endif;
if resinfo% band 4 =4 then
	Print("WaveMark divider disabled.\n")
endif;
if resinfo% band 8 =8 then
	Print("Dummy channels disabled.\n")
endif;
if resinfo[6] then
	Print("Burst Mode sampling:\tON");
endif;
var nch%,lst%[2],c%,err,err$;
View(tvh%);
nch%:=ChanList(lst%[],1+16+16384);	                   'count waveform and wavemark channels, not duplicates
if nch%>0 then
	var chlst%[nch%+1];	                               'size array
	ChanList(chlst%,1+16+16384);	                      'fill it
	View(txtvh%).Print("\nActual v Requested sampling rates:\n  #\tHz\tErr %\n");	'header for table showing match between actual and ideal sampling rates
	for i%:=1 to nch% do
		c%:=chlst%[i%];
		if SampleChanInfo(c%,4)=SampleChanInfo(c%,3) then	'perfect match
			err$:="OK";
		else	                                           'calculate mis-match
			err:=(SampleChanInfo(c%,3)-SampleChanInfo(c%,4))*100.0/SampleChanInfo(c%,3);	'% mismatch between actual and requested sample rates
			err$:=Print$("% .2f%%",err);
		endif;
		View(txtvh%).Print("%3d\t%.1f\t%s\n",c%,SampleChanInfo(c%,4),err$);	'print result
	next;
endif;
return 1;
end;

func ModeTab%()	                                     'continuous/timed/triggered (based on code from SG)
var mode%;
Var i%;
Var list%;
Var code%;

View(txtvh%);
Print("\nSAMPLING MODE:\n");
mode%:=SampleMode(0);	                               'Get the sampling mode
docase	                                              'Print what it is
case mode%=1 then
	Print("Continuous\n");				
case mode%=2 then
	Print("Timed\n");
	Print("For\t%d s\n", SampleMode(-1));	             'Print the times
	Print("Every\t%d s\n", SampleMode(-2));	          'Does not tell us whether minutes are used
case Mode%=3 then
	Print("Triggered:\n");
	Print("For\t%d s...\n\n", SampleMode(-1));	       'For in triggered mode doesn-t make sense anymore
	Print("Trigger channel:\t%d\n", Samplemode(-3));
	for i% := 1 to 4 do
		if SampleTrigger(i%,-1) > 0 then
			Print("Trigger %d set-up\n",i%);			
			Print("Trigger channel:\t%d\n",SampleTrigger(i%,-1));
			code%:=SampleTrigger(i%,-2);
			if code% =-1 then
				Print("Trigger code:\tAccept all\n");
			else
				Print("Trigger code:\t%s %02x\n",Chr$(SampleTrigger(i%,-2)),SampleTrigger(i%,-2));	
			endif
			Print("Start time:\t%f s\n",SampleTrigger(i%,-3));
			Print("End time:  \t%f s\n", SampleTrigger(i%,-4));
			list%:=SampleTrigger(i%,-5);
			docase
			case list%=-1 then
				Print("All channels\n\n");
			case list%=-1 then
				Print("One channel\n\n");
			case list%=-1 then
				Print("Multiple channels\n\n");
			endcase
		endif
	next
endcase;
return 1;
end;

func SequencerTab%()	                                  'contents of sequencer tab insofar as we can find out (based on code from SG)
var seq$;
var Seqstep%;

View(txtvh%);
Seq$:=SampleSequencer$();	                            'Find out if we have an output sequencer file
Print("\nOUTPUT SEQUENCER\n");
if seq$="" then
	Print("None or Graphical Sequencer used\n");	       'Can-t find out whether a graphical seq is in use
else
	Print("%s\n", seq$);	                               'Print the name of the file		
	seqstep%:=SampleSeqCtrl(1);	                      'How is the sequencer controlled
	Print("Sequencer controlled by:\t");
	docase
	case seqstep%= 0 then
		Print(" Keyboard, control panel and script\n");
	case seqstep%= 1 then
		Print(" Control panel and script\n");
	case seqstep%= 2 then
		Print(" Script only\n");
	endcase;
	ok%:=FileOpen(seq$,2);	                            'try to open the sequence file
	if ok%<0 then
		Print("WARNING: Sequence file was not found in this folder.\n");
	else
		FileClose(0,-1);	                               'close the sequence file
	endif;
endif;
return 1;
end;

Func PlaywaveTab%();	                                  'contents of the playWave tab  (based on code from SG)
Var list$;
var key$;
var i%,k%;
var num%;
var trig%;

View(txtvh%);
list$:=PlayWaveInfo$();	                               'list of codes that trigger arbitrary outputs
if list$ > "" then	                                  'if we found some then
	Print("\n\PLAYWAVES\n#\tName\tKey\tDAC(s)\tRate(Hz)\tLinked to\tTriggered\tSpeed\tCycles\n");	'find out everrything we can	
	for k%:= 1 to Len(list$) do;
		key$:=Left$(list$,1);	
		num%:=PlayWaveChans(key$);
		var ch%[num%],ch$;
		PlayWaveChans(key$,ch%[]);
		for i%:=0 to num%-1 do
			ch$+=Print$("%d ",ch%[i%]);	                'encode channel list as a string for printout
		next;
		Print("%d\t%s\t %s\t%s\t%.2f\t", k%,PlaywaveLabel$(key$),key$,ch$,PlayWaveRate(key$));		
		If PlayWaveLink$(key$)>"" then
			Print("%s\t", PlayWaveLink$(key$));
		else
			Print("   -\t");			
		endif
		trig%:=PlaywaveTrigger(key$);
		docase
		case trig%=0 then
			Print("  no\t");
		case Trig%=1 then
			Print("  yes\t");
		case Trig%=-1 then
			Print("  error\t");
		endcase;		
		Print(" x%2.1f\t",PlaywaveSpeed(key$));
		Print("%3d\n", PlayWaveCycles(key$));
		list$:=DelStr$(list$,1,1);	                      'delete this key from the list
		ch$:="";	                                        'reset
	next;
endif;
return 1
end;

func Automation%()	                                  'Print the automation tab
var fpath$,fname$,tcommit%,tcommit$,a$,sst%;

View(txtvh%);
Print("\nAUTOMATION\n");
Print("Automatic file name generation\n");
fpath$:=FilePath$(3);
if Len(fpath$)=0 then fpath$:="None"; endif;
Print("Auto file path:\t%s\n", fpath$);
fname$:=SampleAutoName$();
if Len(fname$)>0 then	                               'filename template set
	Print("File name template:\t%s\n",fname$);	       'print name
	Print("Repeat count:\t%d\n",SampleRepeats());	    'repeat count not applicable if no name template
endif;

Print("\nData security\n");
tcommit%:=SampleAutoCommit();
if tcommit%=0 then
	tcommit$:="off";
else
	tcommit$:=Print$("%.3d s",tcommit%);
endif;
Print("Automatic flush to disk every:\t%s\n",tcommit$);
if SampleAutoComment()then a$:="on"; else a$:="off"; endif;
Print("Comment file at end of sampling:\t%s\n",a$);
if SampleAutoFile()then a$:="on"; else a$:="off"; endif;
Print("Save file at end of sampling: \t%s\n",a$);

Print("\nStarting and Stopping\n");
sst%:=SampleStartTrigger();
docase
case sst%=-1 then a$:="use previous";
case sst%=0 then a$:="not triggered";
case sst%=1 then a$:="first file triggered";
case sst%=2 then a$:="all files triggered";	
endcase;
Print("Triggering:\t%s\n",a$);
if SampleLimitSize()<0 then
	a$:="off";
else 
	a$:=Print$("%4d kB",SampleLimitSize());
endif;
Print("Stop at size:\t%s\n",a$);
if SampleLimitTime()<0 then
	a$:="off";
else
	a$:=Print$("%4d s",SampleLimitTime());
endif;
Print("Stop at time:\t%s\n",a$);
return 1;
end

proc HrMinSec(t,&hr%,&mins%,&s%);	                   't in sec returned in hr:min:s format
s%:=t mod 60.0;
mins%:=t/60.0;
if mins% > 59 then
	hr%:=mins%/60;
	mins%:=mins% mod 60;
endif;
return;
end;

func ActiveCursors%()
var i%,n%,k%;
var nc%;	                                              'number of cursors found
var c%[10];	                                           'array string cursor status (exists: 1 or does not exist: 0)
var CAparam[11],item%,val$[10];	                      'active cursor parameters
var ACmode$[21];	                                     'active cursor modes
ACmode$[0]:="     Static"; ACmode$[1]:="     Maximum value"; ACmode$[2]:="      Minimum value"; 
ACmode$[3]:="  Maximum excursion"; ACmode$[4]:="          Peak find"; ACmode$[5]:="        Trough find"; 
ACmode$[6]:="          Threshold"; ACmode$[7]:="   Rising threshold"; ACmode$[8]:="  Falling threshold"; 
ACmode$[9]:="    Steepest rising"; ACmode$[10]:="   Steepest falling"; ACmode$[11]:="Steepest slope (+/-)"; 
ACmode$[12]:="         Slope peak"; ACmode$[13]:="       Slope trough"; ACmode$[14]:="        Data points"; 
ACmode$[15]:="    Repolarisation%"; ACmode$[16]:="         Expression"; ACmode$[17]:="      Turning point"; 
ACmode$[18]:="             Slope%"; ACmode$[19]:=" Outside thresholds"; ACmode$[20]:="  Inside thresholds";

var pr$[11];
pr$[0]:="       Search Mode"; pr$[1]:="Channel"; pr$[2]:= "Title"; pr$[3]:="Start"; 
pr$[4]:="End"; pr$[5]:="Default"; pr$[6]:="Ampl/Thr";	 '/Threshold/Baseline
pr$[7]:="Hysteresis";	                               'percent
pr$[8]:="Width";  
pr$[9]:="T 100%";

View(txtvh%).Print("\n\nACTIVE CURSOR SEARCHES\n");
View(tvh%);
for i%:=0 to 9 do
	c%[i%]:=CursorExists(i%);
next;
nc%:=ArrSum(c%);
if nc%=1 and CursorActive(0) =0 then	                'if only cursor 0 and it is static 
	View(txtvh%);
	Print("No active Cursor searches\n");
else
	View(txtvh%);
	Print(" Cursor\t");
	for i%:=0 to 9 do
		Print("%s\t",pr$[i%]);
		if i%=0 then Print("\t"); endif;	                'extra space for long mode names
	next;
	Print("\n\t\t\t\t\t\t(Expr)\t\t(Minstep)\tB.line/pts\t Repol%\t\tThr. 2\n");
	View(tvh%);
	for i%:=0 to 9 do	                                  'which cursors exist?
		if c%[i%]=1 then
			c%[i%]:=i%;	                                  'replace with cursor number
		endif;
	next;
	ArrSort(c%,1);	                                     'sort descending: moves existing cursors to the front
	ArrSort(c%[:nc%]);	                               'convert to ascending list of existing cursors
	for i%:=0 to nc%-1 do	                            'get parameters of each cursor
		ArrConst(CAparam[],0);	                         'initialise arrays
		for k%:=1 to Len(val$[])-1 do
			val$[k%]:="";
		next;
		CAparam[0]:=CursorActive(c%[i%]);
		if CAparam[0]>0 then	                            'if cursor is active then get the parameters
			View(txtvh%).Print("\n%3d\t%s\t",c%[i%],ACmode$[CAparam[0]]);
			for item%:= 1 to 10 do
				if item%=4 and c%[i%]=0 then
					CAparam[item%]:=CursorActiveGet(c%[i%],8,val$[7]);	'fudge to get C0 minstep into the same column as default position
				else					
					CAparam[item%]:=CursorActiveGet(c%[i%],item%,val$[item%-1]);
				endif;
				if item% <> 8 then	                      'no column needed for min step
					repeat
						n%:=InStr(val$[item%-1],"Cursor(");
						if n%>0 then
							val$[item%-1]:=DelStr$(val$[item%-1],n%+8,1);	'delete ) after the cursor number
							val$[item%-1]:=DelStr$(val$[item%-1],n%+1,6);	'abbreviate <Cursor> to <C>
						endif;
						n%:=InStr(val$[item%-1],"MaxTime()");
						if n%>0 then
							val$[item%-1]:=DelStr$(val$[item%-1],n%+4,5);	'abbreviate MaxTime() to MaxT
						endif;						
					until n%=0;
					docase	                               'if no string parameter print return value of CursorActiveGet()
					case val$[item%-1]="" and item%=1 then 
						View(txtvh%).Print("\t%3d\t%s",CAparam[item%],View(tvh%).ChanTitle$(CAparam[item%]));	'channel number as an integer and add title
					case c%[i%]=0 and item%=4 then
						View(txtvh%).Print("\t%.3f",CAparam[item%]);	
					case val$[item%-1]="" and item%<> 4 then
						View(txtvh%).Print("\t%.3f",CAparam[item%]);	'other numbers as floating point
					case item% = 9 and CAparam[0]<>15 then 
						'do nothing: T 100% column only relevant for <Repolarisation %>
					case item% = 10 and CAparam[0]<19 then	 'nothing to print
						'do nothing: thr 2 only valid for modes 19 and 20
					else
						View(txtvh%).Print("\t%s",val$[item%-1]);	'print sting parameter
					endcase;
				endif;
			next;
		endif;
	next;	
endif;
View(txtvh%).Print("\n");
return 1;
end;

func ChannelProcesses%()
var lst%[2],nch%,ndel%,i%,j%,p%;
var chttl$,np%,arg,arg2;
var proc$[10];	                                        'list of processes 
proc$[0]:="Rectify"; proc$[1]:="Smooth"; proc$[2]:="DC Remove"; proc$[3]:="Slope"; proc$[4]:="Time shift"; 
proc$[5]:="Down sample"; proc$[6]:="Interpol."; proc$[7]:="Chan match"; proc$[8]:="RMS ampl."; proc$[9]:="Median filt";
var pr$[10];	                                        'labels for process arguements
pr$[1]:="Range (s)"; pr$[2]:="Range (s)"; pr$[3]:="Range (s)"; 
pr$[4]:="Time shift"; pr$[5]:="Ratio"; pr$[6]:="Interval"; 
pr$[7]:="Chan match"; pr$[8]:="Range (s)"; pr$[9]:="Range (s)"; 

View(txtvh%).Print("\nONLINE CHANNEL PROCESSES\n");
View(tvh%);
nch%:=ChanList(lst%,1+512);
var chlst%[nch%+1];
ChanList(chlst%,1+512);	                               'get list of waveform channels
for i%:=1 to nch% do
	if ChanProcessInfo(chlst%[i%])= 0 then	             'no processes on this channel
		chlst%[i%]:=-1;
		ndel%+=1;	                                     'count of number removed
	endif;
next;
if nch%-ndel%=0 then
	View(txtvh%).Print("No online channel processes\n");
	return 1;
endif;
ArrSort(chlst%[1:],1);	                               'sort descending, i.e., unprocessed channels to the end
chlst%[0]:=nch%-ndel%;
ArrSort(chlst%[1:chlst%[0]]);	                         'sort channels with processes back into ascending order
for i%:=1 to chlst%[0] do
	chttl$:=ChanTitle$(chlst%[i%]);
	np%:=ChanProcessInfo(chlst%[i%]);
	View(txtvh%).Print("Channel:\t%4d\t%s\tProcesses:\t%d\n",chlst%[i%],chttl$,np%);
	'	View(txtvh%).Print("-----------------------------------------------------------\n");
	for j%:=1 to ChanProcessInfo(chlst%[i%]) do	       'for each process do
		p%:=ChanProcessInfo(chlst%[i%],j%);	             'process index
		View(txtvh%).Print(" Process:\t%4d\t%s",j%,proc$[p%]);	'print process type
		arg:=ChanProcessArg(chlst%[i%],j%,1);	          'first channel process arguement
		docase
		case p%=0 then
			View(txtvh%).Print("\t    -\n");	             'rectify has no arguements
		case p%=6 then
			arg2:=ChanProcessArg(chlst%[i%],j%,2);	       'interpolate has two arguements
			View(txtvh%).Print("\t%s:\t%.4f\tAlignment\t%.4f\n",pr$[p%],arg,arg2);
		else
			View(txtvh%).Print("\t%s:\t%.4f\n",pr$[p%],arg);	'all other processes have one
		endcase;		
	next;
	View(txtvh%).Print("\n");	
next;
return 1;
end;

func Conditioners%()
var lst%[2],nch%,ndel%,i%,j%,ok%,ttl$,n%,u$;
var gain;
var type$;
var ac$[2],notch$[2];
ac$[0]:="  off"; ac$[1]:="   on";
notch$[0]:="  off"; notch$[1]:="   on";
var floname$,fhiname$;	                               'name of low and hi pass filters


View(tvh%);
nch%:=ChanList(lst%,1+16+16384);	                      'waveform and Wavemark channels
if nch%=0 then	                                        'no sampled waveform or wavemark channels found
	return 1;
else
	View(txtvh%).Print("\nSIGNAL CONDITIONER(S)\n");
endif;
var chlst%[nch%+1],plst%[nch%+1];	                   'arrays for channel and corresponding port numbers
ChanList(chlst%[],1+16+16384);
for i%:=1 to chlst%[0] do
	plst%[i%]:=ChanPort(chlst%[i%]);	                   'generate a list of ports in use
	if plst%[i%]<0 then
		ndel%+=1;
	endif;
next;
ArrSort(plst%[1:],1,chlst%[1:]);	                      'sort descending: chans with no port sort to the end
chlst%[0]-=ndel%;
plst%[0]:=chlst%[0];
ArrSort(plst%[1:plst%[0]],0,chlst%[1:]);	             'ascending list of existing ports and chans

var ctype[chlst%[0]+1];
ctype[0]:=chlst%[0];
for i%:=1 to ctype[0] do
	ctype[i%]:=CondType(plst%[i%]);	                   'list conditioner type linked to each port
next;

if ArrSum(ctype[1:])=0 then
	View(txtvh%).Print("No signal conditioners found.\n");
else
	View(txtvh%).Print("Type\tChannel\tTitle\tPort\tGain\tOffset\tInput signal\tHighPass\tLowPass\tNotch\tAC couple\t     HP type\t     LP type\t\n\t\t\t\t\t\t\t\t Cut (Hz)\t Cut (Hz)\n");
	var in%,offs,low,hi,notch%,ac%,typeL%,typeH%,low$,hi$,src$[100];
	for i%:=1 to chlst%[0] do
		ttl$:=ChanTitle$(chlst%[i%]);
		if ctype[i%]=0 then
			'ignore ports with no signal conditioner
		else
			type$:=CondRevision$(plst%[i%]);
			n%:=InStr(type$,"CYBERAMP ");
			if n% > 0 then
				type$:="CA"+Mid$(type$,n%+9,3);	          'abbreviate cyberamp entry for tabulation
			endif;
			n%:=InStr(type$,"1902");
			if n% > 0 then
				type$:="1902/"+Mid$(type$,n%+4);	          'separate model and revision nr.
			endif;
			gain:=CondGain(plst%[i%]);
			View(txtvh%).Print("%s\t%3d\t%s\t%3d\tx%4g\t",type$,chlst%[i%],ttl$,plst%[i%],gain);
			if ctype[i%]<>3 then	                         'get 1902 / cyberamp info
				ok%:=CondGet(plst%[i%],in%,gain,offs,low,hi,notch%,ac%,typeL%,typeH%);
				CondFilterType(plst%[i%],0,typeL%,floname$);	'get name of lo pass filter
				CondFilterType(plst%[i%],1,typeH%,fhiname$);	'get name of hi pass filter
				LSpacePad(12,floname$);	                   'minimum length of names for correct tabulation
				LSpacePad(12,fhiname$);
				if low=0 then low$:="All Pass"; else low$:=Print$("% .1f",low); endif;
				if hi=0 then hi$:="  DC"; else hi$:=Print$("% .1f",hi); endif;
				CondSourceList(plst%[i%],src$[]);
				LSpacePad(12,src$[in%]);
				View(txtvh%).Print("%.4g %s\t%s\t%s\t%s\t%s\t%s\t%s\t%s\n",offs,u$,src$[in%],hi$,low$,notch$[notch%],ac$[ac%],fhiname$,floname$);
			endif;
		endif;
	next;	
endif;

View(txtvh%);
var nf%;	                                              'nr. of special features
var f%;	                                              'index of special feature	
var fname$,flow,fhigh,flags%,fsel%;
var hdr%:=0;

for i%:=1 to plst%[0] do	                            'process all ports with a conditioner attached	
	if ctype[i%]=1 then	                               'only 1902s have <features>
		nf%:=CondFeature(plst%[i%]);	                   'number of special features available
		if nf% > 0 then	                               'special features detected
			if hdr%=0 then	                               'print table header if we found at leat one channel with features enabled
				Print("\nSIGNAL CONDITIONER FEATURES\n");
				Print("Port\tChannel\tTitle\t     Feature\tSetting\tLow\tHigh\n");
			endif;
			Print("\n%3d\t%3d\t%s\t",plst%[i%],chlst%[i%],View(Tvh%).ChanTitle$(chlst%[i%]));	'basic info of next channel on a new line
			for j%:=1 to nf% do
				f%:=CondFeature(plst%[i%],j%,fname$);	    'names of current features and number of possible discrete values (or 0 meaning continuously variable)
				if f%=0 then	                            'feature is continously variable
					CondFeature(plst%[i%],j%,fname$,flags%,flow,fhigh);	'get limits of range
					LSpacePad(12,fname$);	                'pad feature name to at least 12 characters
					Print("%s\t\t% .4g\t% .4g\n");
				else	                                     'discrete settings
					var flist$[f%];
					CondFeature(plst%[i%],j%,fname$,flags%,flist$[]);	'get list of discrete settings
					fsel%:=CondFeature(plst%[i%],j%);
					Print("%s\t%s\n",fname$,flist$[fsel%]);	'print details of current selection
				endif;
				Print("\t\t\t");	                         'align all features of the same channel
			next;
		endif;
	endif;		
next;
return 1;
end;

proc LSpacePad(ln%,&src$)	                            'pad a string <src$> with leading spaces to achieve a minimum length of Ln%
var n%,i%;

n%:=ln%-Len(src$);
if n%>0 then
	for i%:=1 to n% do
		src$:=" "+src$;	                               'extend name to required minimum length
	next;
endif;
return;
end;

which you can select the components of the sampling configuration that you want 
to include in the printed report. 

 This script requires that you have a 1401 connected and switched on, as it needs 
to generate a data file in order to retrieve the information from the current 
configuration. 

Scripts: Signal  
Q. I have a series of data files imported from an in-house recording system that I 

would like to analyse. The problem I have is that our in-house system stores data 
in columns, which when opened in Signal shows a single frame with multiple 
channels. I would like each channel to be displayed as an individual frame in 
Signal. Is there any way to do this?  

A. The attached script, ChansIntoFrames.sgs, will create a new multiple frame 
file, with each channel from the selected source file assigned to a separate frame 
of data. 

Did you know…? 
When importing WAV files to Spike2, the data file is no longer limited to use an integer 
multiple of microseconds as the timebase. The importer now reads the WAV file header 
and creates a data file at exactly the same sample rate as set in the original WAV file. 

Recent questions   
Q. I would like to setup a PSTH result view where data is only included if the value 

on a waveform channel (a sine wave which is proportional to a stimulus I 
delivered in the experiment) is within a given voltage range. 

A. You can set up a PSTH to only include sweeps of data that occur when a 
waveform channel is within a range of values by using the auxiliary 
measurements option in the PSTH dialog. Set the Auxiliary channel to be the 
waveform channel you want to monitor, set the Accept field to Inside range and 
then the Range fields to the start and end values that meet your criteria. Each 
sweep that occurs is then only added to the result if it matches the values set. 

 
Settings for PSTH with auxiliary channel setup 

User group      
If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 
 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$ChansIntoFrames|Creates a new data file with each channel of the source file assigned to a seperate frame

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 11/07/08

var data%;          'View handle for the data file
var chans%[100];    'Available channels in a file
var title$;            'Title of the data file
var frameFile%;     'New file to hold frames
var c%;                 'Counter for the frames

if ViewKind() = 0 then
data%:= View();		'Make sure we have a data file open
FrontView(data%);       'bring it to the front
else
Message("File is not a data file!\nOpen a data file and then run the script");	'Otherwise warn user and halt script
halt;endif;

title$:= WindowTitle$();    'Get title of source file

ChanList(chans%[],1);       'Get a list of all channels

frameFile%:=SetCopy(2,title$,0);    'Create a copy of the file

for c%:=1 to chans%[0] do           'Cycle through the channels
    Frame(c%);                          'Get current frame
    FrontView(frameFile%);            'Switch to new file
    ArrConst(View(frameFile%,1).[],View(data%,c%).[]);          'Copy the data from current channel to frame
    AppendFrame();          'Create a new frame for the next channel
next;

